
Odhad hierarchické sm¥si

Hierarchická sm¥s je p°esn¥ to, co se pod pojmem �hierarchická� myslí. Naho°e je sm¥s s n¥ko-
lika komponentami. Kaºdá komponenta je op¥t sm¥sí a m·ºe ukazovat na dal²í sm¥si jako své
komponenty.

Hlavní význam takového modelu je z°ejm¥ v tom, ºe celý problém (klastrování a klasi�kace) je
na nejvy²²í úrovni zhruba rozd¥len do n¥kolika £ástí (komponenty vrchní sm¥si). Kaºdá z t¥chto
£ástí je potom dále podrobn¥ji zkoumána (d¥lena) podle dal²ích podrobností v komponentách
p°íslu²né sub-komponenty z vy²²í £ásti. Pro lep²í p°edstavu uvedeme konkrétní p°íklad.

P°íklad

Sledujeme jízdu osobního automobilu a rádi bychom detekovali, zda jde o jízdu ekonomickou
(opatrnou) nebo sportovní (divokou). To, jak se jízda projevuje se nejlépe pozná z m¥°eného
p°í£ného a podélného zrychlení. Ekonomická jízda má tato zrychlení v¥t²inou nulové, jen vy-
jíme£n¥ se objevují malé odchylky. Divoká jízda naopak má tato zrychlení v¥t²inou nenulová,
ob£as mohou klesnout sm¥rem k nule ale ob£as také mohou vylet¥t do v¥t²ích hodnot.

Ale to, jak se zm¥ny v akceleraci projeví bude z°ejm¥ záviset na tom, kde se automobil zrovna
pohybuje. Vytipujeme si 3 druhy prost°edí: ve m¥st¥, mimo m¥sto a na dálnici.

Vrchní komponenta modelu se bude týkat prost°edí. Bude mít 3 komponenty (podle typ· prost°edí)
a jako data zvolíme rychlost (do 50 km/h m¥sto, mezi 50 a 100 km/k mimo m¥sto a nad 100
km/h dálnice - tyto hodnoty vyuºijeme pro apriorní nastavení horní sm¥si).

Kaºdá z práv¥ jmenovaných komponent bude zase sm¥sí s dv¥ma komponentami. Jedna pro
ekonomickou jízdu (malá zrychlení) a druhá pro divokou jízdu (v¥t²í zrychlení).

Model

Obecná formulace hierarchického modelu sm¥si plyne z uvedeného p°íkladu. Pro jednoduchost
se ale budeme i nadále p°íkladu drºet. Nejprve zavedeme zna£ení:

Horní sm¥s

yt jsou jsou data pro horní sm¥s (rychlost)

ct je ukazovátko pro horní sm¥s

Θ jsou parametry horních komponent (závisí na ct)

α je parametr horního ukazovátka

Dolní sm¥si

dt jsou data pro dolní sm¥si (p°í£né a podélné zrychlení)

kt je ukazovátko pro dolní sm¥si (kt bude závislé na ct)

ϑ jsou parametry dolních komponent (závisí na ct i kt)

β jsou parametry dolních ukazovátek

Model pak bude

f (yt, dt|ct, kt,Θ, ϑ, α, β) = f (yt|ct,Θ, α) f (dt|kt, ct, ϑ, β)

kde první £ást na pravé stran¥ je model horní sm¥si a druhá £ást jsou modely dolních sm¥sí.

Odhad parametr· modelu

Pro odvození odhadu vyjdeme op¥t ze sdruºené pravd¥podobnosti J (kdeDt−1 = [y (t− 1) , d (t− 1)]
je mnoºina v²ech starých dat)

J = f (yt, dt, ct, kt,Θ, ϑ, α, β|Dt−1) =

= f (yt|ct, Θ) f (Θ|y (t− 1))︸ ︷︷ ︸
horní komponenty

× f (ct|α) f (ct−1|y (t− 1)) f (α|y (t− 1))︸ ︷︷ ︸
horní pointer model

×

× f (dt|ct, kt, ϑ) f (ϑ|d (t− 1))︸ ︷︷ ︸
dolní komponenty

× f (kt|ct, β) f (β|d (t− 1))︸ ︷︷ ︸
dolní pointer model

.

Ozna£íme:

� predikce z horních komponent

pct =

ˆ
Θ∗
f (yt|ct,Θ) f (Θ|y (t− 1)) dΘ

.
= f

(
yt|ct, Θ̂t−1

)
� predikce z dolních komponent

qkt|ct =

ˆ
ϑ∗
f (dt|kt, ct, ϑ) f (ϑ|d (t− 1)) dϑ

.
= f

(
dt|kt, ct, ϑ̂t−1

)
� predikce horního pointru

α̂ct|ct−1
=

ˆ
α∗
f (ct|ct−1, α) f (α|y (t− 1)) dα

� predikce dolního pointru

β̂kt|ct =

ˆ
β∗
f (kt|ct, β) f (β|d (t− 1)) dβ

� staré váhy
fct−1

= f (ct−1|y (t− 1))

Sdruºenou predikci v²ech ukazovátek m·ºeme psát

P = f (ct, ct−1, kt|y (t) , d (t)) ∝ pct α̂ct|ct−1
fct−1 × qkt|ct β̂kt|ct

kde prvá £ást (do ×) pat°í horní sm¥si a druhá £ást dolním sm¥sím.

Sdruºená pravd¥podobnost ct a ct−1 je

Wt = f (ct, ct−1|y (t)) =
∑
kt

P,

pravd¥podobnost ct je marginála Wt

wu;t =
∑
ct−1

Wt

a pravd¥podobnost kt je
wl;t =

∑
ct

∑
ct−1

P.

Tím je ur£en výpo£et pot°ebných vah.

2



P°epo£et statistik

P°epo£et statistik pro ukazovátka je standardní p°epo£et pro diskrétní model s p°íslu²nými
vahami komponent.

P°epo£et statistik pro komponenty závisí na modelu, který se pro n¥ pouºije. Zde se op°eme o
jiº zmín¥ný p°íklad o sledování jízdy osobního automobilu.

Horní komponenty pracují s veli£inou rychlost a o£ekáváme centra komponent v hodnotách 40
(m¥sto), 80 (mino m¥sto) a 110 (dálnice). I kdyº se jedná o nezápornou veli£inu, m·ºeme pro
ní uvaºovat normální model (data budou p°ibliºn¥ symetrická podle zmín¥ných center).

Dolní komponenty modelují zrychlení. Protoºe o£ekáváme, ºe rozjezd· bude p°ibliºn¥ stejn¥ jako
brzd¥ní a zatá£ek doleva tolik, kolik zatá£ek doprava, £ekáme, ºe data budou symetrická kolem
po£átku - p°i£emº maximální po£et hodnot pro ekonomickou jízdu bude v nule, pro sportovní
jízdu kolem nuly. Klastry tedy budou v nule (ekonomická jízda) a jakýsi prstenec kolem nuly
(sportovní jízda). Takové klastry ale nejsou vhodné pro detekci pomocí hustot pravd¥podobností.
Proto tyto data p°evedeme na absolutní hodnoty a dále s nimi budeme po£ítat v této form¥.

Máme tedy nezáporná data: v jedné komponent¥ nahromad¥ná p°edev²ím na nule, v druhé
kousek od nuly. Modely t¥chto komponent hledáme tedy mezi nezápornými rozd¥leními - viz
úloha ODHAD SM�SI PRO NEZÁPORNÁ DATA.

Program

// Estimation of hierarchical mixture

// - upper - SPEED (with three normal components)

// - lower - SPEED DIFFERENCE and CROSS ACCELERATION

// (economical and sport-like driving is to be detected)

// absolute values of data -> first quadrant

// two lower components: 1. HNm, 2. Exp

// ----------------------------------------------------------

exec ScIntro.sce, mode(0)

nd=1500; // number of data for estimation

ni=50; // virtual length of initialization

load dataOkruhy.dat; // real data from a car

dat=dataOkruhy.dataOrig(:,3001:10:$);

dat=dat(:,3601:7240);

// data

v=dat(2,1:nd); // upper data - velocity

d=abs([diff(v) 0]); // abs. values of velocity variations

a=20*abs(dat(9,1:nd)); // abs. values of cross-acceleration

// variables

yt=v; // data -> varibles (speed)

dt=[d;a]+1e-8; // data -> varibles (accelerations)

// INITIAL PARAMETERS ============================================

thC0=[50 70 100]; // speeds in: town, out-town, motorway

thP0=[.2 .5 .3]; // probabilities of going in -"-

nc0=length(thC0); // number of upper components

nv=2; // number of lower variables

nc=[2 2 2]; // number of lower components

w0=fnorm(ones(1,nc0)); // initial upper weights

w=list();

for i=1:nc0 // initial lower weights

w(i)=3+rand(1,nc(i));

w(i)=w(i)/sum(w(i));

end

thCr=list(); thCe=list();

// parameters of lower components

// - the data are: - d1 long. acceleration (mean value is 2.6

// - d2 cross accel. (mean after normaliz. is 3.8

// d1 d2

thCe(1)= [1 .2]'; // Exp (1st comp. of 1st mixture)

thCr(1)=%pi/2*[8 4]'; // HNm (2nd komp. of 1st mixture)

thCe(2)= [1 .2]'; // 2nd mixture

thCr(2)=%pi/2*[8 4]'; // expectation -> statistics

thCe(3)= [1 .2]'; // 3rd mixture

thCr(3)=%pi/2*[8 4]';

// INITIALIZATION ================================================

// UPPER mixture

for j=1:nc0

Ps=[thC0(:,j);1]; // Regression vector

Cyp.V=Ps*Ps';

Est0.Cy(j)=Cyp; // information matrix

Est0.Cy(j).th=thC0(:,j); // parameters

Est0.Cy(j).sd=.1; // std of noise

end

Est0.ka=1*rand(1,nc0,'u'); // counter

Est0.Cp.V=rand(nc0,nc0,'u')+1;// statistics for pointer estimation

Est0.Cp.th=fnorm(Est0.Cp.V,2);// parameters of pointer model

Est0.ct=1; // initial value of the pointer

Est0.nc=nc0; // number of upper components

Est0.w=w0; // initial weight

Est0.W=zeros(nc0,nc0);

Est0.tht=zeros(nc0,nd);

Est0.wt=zeros(nc0,nd);

// LOWER mixture

for i=1:nc0

Esti.i=i; // label of lower component

Esti.ka=ni*ones(1,2); // counter

// first component

Esti.Cy(1).V=ni*thCe(i); // statistics

Esti.Cy(1).th=thCe(i); // initial parameters

// second component

Esti.Cy(2).V=ni*thCr(i); // statistics

Esti.Cy(2).th=.8*thCr(i); // initial parameters

Esti.Cp.V=rand(1,nc(i),'u')+1; // pointer statistics

Esti.Cp.th=fnorm(Esti.Cp.V); // pointer parameters

Esti.nc=nc(i); // number of lower components

Esti.w=w(i); // initial weights

Esti.ct=zeros(1,nd);

Est(i)=Esti; // lower component construction

end

//pause

// ESTIMATION loop ===============================================

tic();

kk=ceil(nd/10); printf('\n 2 4 6 8 |\n ')

for t=1:nd

if t/kk==fix(t/kk), printf('.'); end

[Est0,Est]=MEst78(yt(:,t),dt(:,t),Est0,Est);

[xxx,Est0.ct(1,t)]=max(Est0.w);

Est0.wt(:,t)=Est0.w';

Est0.tht(:,t)=[Est0.Cy(1).th,Est0.Cy(2).th,Est0.Cy(3).th]';

for i=1:nc0

Est(i).tht(:,t)=[Est(i).Cy(1).th(:);Est(i).Cy(2).th(:)];

[xxx,Est(i).ct(t)]=max(Est(i).w);

Est(i).wt(:,t)=Est(i).w';

end

end

// end of estimation loop ========================================

disp ' ', time=toc()/60; printf(' time = %5.2f min.\n',time)

// RESULTS

d=list();

for i=1:3

m=find(Est0.ct==i);

d(i)=dt(:,m);

for j=1:2

k=find(Est(i).ct(m)==j);

Ct.t=d(i)(:,k);

E(i).C(j)=Ct; // data in components

end

end

// data clusters colored with respect to components

tx=['in town';'outside town';'motorway'];

set(scf(112),'position',[400 150 1000 350]);

for i=1:3

subplot(1,3,i)

plot(E(i).C(1).t(1,:),E(i).C(1).t(2,:),'b.','markersize',3)

plot(E(i).C(2).t(1,:),E(i).C(2).t(2,:),'r.','markersize',3)

title(tx(i))

set(gca(),'data_bounds',[0 30 0 20]);

legend('economic driving','sport driving');

end

Procedura

function [Eh,Ed]=MEst78(y,d,Eh,Ed)

// estimation of both upper and lower mixture

// upper: 3 norm. components

// lower: 2 comp. - first half-normal

// - second exponential

nc=Eh.nc;

al=Eh.Cp.th;

Wh=Eh.w;

ny=length(d);

wd=list();

v=list();

// PROXIMITIES AND WEIGHTS

for i=1:nc

// PROXIMITY UPPER

[xxx,Gh(i)]=GaussN(y,Eh.Cy(i).th,Eh.Cy(i).sd); // Gauss

// PROXIMITY LOWER

[xxx,Gd(1)]=HnormN(d,Ed(i).Cy(1).th) // (th is "s"; E=.8*s)

[xxx,Gd(2)]=ExpN(d,Ed(i).Cy(2).th); // (th is expectation)

Ld=Gd-max(Gd);

q(i,:)=fnorm(exp(Ld))'; // proximity lower

wd(i)=Ed(i).w;

be(i,:)=Ed(i).Cp.th;

end

Lh=Gh-max(Gh);

p=fnorm(exp(Lh))'; // proximity upper

W=fnorm((Wh'*p).*al); // joint pf of c_t and c_t-1

Wh=sum(W,1); // uppwer weight

[xxx,ch]=max(Wh);

Eh.w=Wh;

Wd=fnorm((Wh'*ones(1,2)).*(q.*be)); // lower weight

for i=1:Eh.nc

Ed(i).w=Wd(i,:);

end // ---------------------------------------- end of proximity

// UPDATE OF COMPONENTS AND COMPUTATION OF POINT ESTIMATES

// UPPER mixture

Ps=[y';1]; // Regression vekcor

for i=1:Eh.nc

Eh.Cy(i).V=Eh.Cy(i).V+Wh(i)*Ps*Ps'; // information matrix

Eh.ka(i)=Eh.ka(i)+Wh(i); // counter

// point estimates of upper parameters

// for scalar y !!!

Vy=Eh.Cy(i).V(1,1); // Vyp - y.y

Vyp=Eh.Cy(i).V($,1); // Vyp - psi.y

Vp=Eh.Cy(i).V($,$); // Vp - psi.psi'

Eh.Cy(i).th=inv(Vp+1e-8*eye(Vp))*Vyp; // estimate of reg. coef.

Eh.Cy(i).tht(:,t)=Eh.Cy(i).th';

if 0 // estimate of covariance

Eh.Cy(i).cv=(Vy-Vyp'*inv(Vp+1e-8*eye(Vp))*Vyp)/Eh.ka(i);

end

end

Eh.Cp.V=Eh.Cp.V+W; // pointer statistics

Eh.Cp.th=fnorm(Eh.Cp.V,2); // estimate of pt. parameters

// LOWER mixture

// update of statistics for hnm

for i=1:Eh.nc

Ed(i).Cy(1).V=Ed(i).Cy(1).V+wd(i)(1)*d^2; // stat. = sum of squares

Ed(i).ka(1)=Ed(i).ka(1)+wd(i)(1); // counter

// point estimates of parameters for hnm

Ed(i).Cy(1).th=sqrt(Ed(i).Cy(1).V/Ed(i).ka(1));

// update of statistics for exp

Ed(i).Cy(2).V=Ed(i).Cy(2).V+wd(i)(2)*d; // stat. = sum

Ed(i).ka(2)=Ed(i).ka(2)+wd(i)(2); // counter

// point estimates of parameters for exp

Ed(i).Cy(2).th=Ed(i).Cy(2).V/Ed(i).ka(2);

Ed(i).Cp.V=Ed(i).Cp.V+wd(i); // pointer statistics

Ed(i).Cp.th=fnorm(Ed(i).Cp.V,2); // pointer parameters

end

endfunction

Popis programu

Program je dosti sloºitý. Proto je realizován pomocí procedury MEst78.sci, která obsahuje algo-
ritmus vlastního odhadu komponent jednotlivých sm¥sí.

Hlavní program obstarává

• nataºení a úpravu dat (absolutní hodnoty akcelerací),

• inicializaci odhadu pro horní a dolní sm¥si,

• prezentaci výsledk·.

V £asové smy£ce (ESTIMATION lopp) se volá funkce MEst78(yt(:,t),dt(:,t),Est0,Est), která ob-
starává on-line b¥h odhadovacího algoritmu.

Je tvo°ena t°emi £ástmi

1. Výpo£et vah w pro jednotlivé komponenty, kde hlavním prvkem jsou tzv. proximity -
tj. hodnoty hustoty pravd¥podobnosti p°íslu²né komponenty s dosazenými existujícími
bodovými odhady parametr· a aktuáln¥ zm¥°enými daty, tj. fc

(
yt|Θ̂c;t−1

)
, kde c =

1, 2, · · · , nc (pro v²echny komponenty).

2. P°epo£et statistik, kde data se p°idávají s p°íslu²nou váhou.

3. Výpo£et bodových odhad·.

Kaºdý z t¥chto krok· se provádí nejd°íve pro horní sm¥s a pak pro jednotlivé dolní sm¥si. Pro-
ximity jsou po£ítány pomocí funkcí GaussN, HNormN a ExpN, a pomocí nich jsou dále po£ítány
váhy Wh a Wd.

V dal²í £ásti jsou p°epo£ítávány statistiky a po£ítány bodové odhady parametr· pro jednot-
livé sm¥si: horní s normálním rozd¥lením a dolní s polovi£ním normálním a exponenciálním
rozd¥lením.

